In[1]:=
```
(*This notebook computes and bounds the infinite
  Euler product of $\mathfrak{A}_{p}$'s in the U^{(r)}-sum.*)
```

In[2]:=
```
SetDirectory[NotebookDirectory[]];
<< gln.m
```

GL(n)pack version 15th April 2015 loading ....Copyright(c) 2002-2006 Kevin A. Broughan GL(n)pack comes with ABSOLUTELY NO WARRANTY. This is free software, and you are welcome to redistribute it under the terms of the GNU Public Licence. For assistance consult kab@waikato.ac.nz .

SetDelayed: Tag UpperTriangularMatrixQ in UpperTriangularMatrixQ [m_] is Protected.

GL(n)PACK loaded.

In[4]:=
```
Schu[n_, d_, nota_ : a] :=
 SchurPolynomial[Table[nota[i], {i, 1, d}], Join[ConstantArray[0, d - 2], {n}]]
```

In[5]:=
```
Syma3 = (a[1] - a[2]) (a[1] - a[3]) (a[2] - a[3]);
```

In[6]:=
```
Symb3 = (b[1] - b[2]) (b[1] - b[3]) (b[2] - b[3]);
```

In[7]:=
```
(*U= psi_{1}(p)/p^{u_{1}}, V= psi_{2}(p)/p^{u_{2}}, Z= 1/p^z*)
```

In[8]:=
```
FF2[d_, g_] := Sum[Schu[M + g, d, b] * V^M, {M, 0, Infinity}]
```

In[9]:=
```
FF1[d_, g_] := Sum[Schu[M + g, d, a] * U^M, {M, 0, Infinity}]
```

In[10]:=
```
FF3[d_, g_] :=
 Schu[g, d, a] * FF2[d, g] + Schu[g, d, b] * FF1[d, g] - Schu[g, d, a] * Schu[g, d, b]
```

In[11]:=
```
InvL[X_, d_, nota_ : a] := Product[(1 - nota[i] X), {i, 1, d}]
```

In[12]:=
```
InvRSL[X_, d_, nota1_ : a, nota2_ : b] :=
 Product[Product[(1 - nota1[i] * nota2[j] X), {i, 1, d}], {j, 1, d}]
```

In[13]:=
```
NormaSum[d_] := Cancel[Sum[FF3[d, g] * Z^g, {g, 0, Infinity}] *
   InvL[U, d, a] * InvL[V, d, b] * InvRSL[Z, d, a, b]]
```

In[14]:=
```
(*PGL(2)*)
```

In[15]:=
```
NormaSum[2]
```

Out[15]=

$$
\begin{aligned}
&1 - U\,V\,a[1]\times b[1] - U\,V\,a[2]\times b[1] + U^2\,V\,a[1]\times a[2]\times b[1] - U\,Z\,a[1]\times a[2]\times b[1] + \\
&U\,V\,Z\,a[1]\times a[2]\,b[1]^2 - U\,V\,a[1]\times b[2] - U\,V\,a[2]\times b[2] + U^2\,V\,a[1]\times a[2]\times b[2] - \\
&U\,Z\,a[1]\times a[2]\times b[2] + U\,V^2\,a[1]\times b[1]\times b[2] - V\,Z\,a[1]\times b[1]\times b[2] + \\
&U\,V\,Z\,a[1]^2\,b[1]\times b[2] + U\,V^2\,a[2]\times b[1]\times b[2] - V\,Z\,a[2]\times b[1]\times b[2] - \\
&U^2\,V^2\,a[1]\times a[2]\times b[1]\times b[2] + 4\,U\,V\,Z\,a[1]\times a[2]\times b[1]\times b[2] - \\
&Z^2\,a[1]\times a[2]\times b[1]\times b[2] - U^2\,V\,Z\,a[1]^2\,a[2]\times b[1]\times b[2] + U\,Z^2\,a[1]^2\,a[2]\times b[1]\times b[2] + \\
&U\,V\,Z\,a[2]^2\,b[1]\times b[2] - U^2\,V\,Z\,a[1]\,a[2]^2\,b[1]\times b[2] + U\,Z^2\,a[1]\,a[2]^2\,b[1]\times b[2] - \\
&U\,V^2\,Z\,a[1]\times a[2]\,b[1]^2\,b[2] + V\,Z^2\,a[1]\times a[2]\,b[1]^2\,b[2] - U\,V\,Z^2\,a[1]^2\,a[2]\,b[1]^2\,b[2] - \\
&U\,V\,Z^2\,a[1]\,a[2]^2\,b[1]^2\,b[2] + U\,V\,Z\,a[1]\times a[2]\,b[2]^2 - U\,V^2\,Z\,a[1]\times a[2]\times b[1]\,b[2]^2 + \\
&V\,Z^2\,a[1]\times a[2]\times b[1]\,b[2]^2 - U\,V\,Z^2\,a[1]^2\,a[2]\times b[1]\,b[2]^2 - \\
&U\,V\,Z^2\,a[1]\,a[2]^2\,b[1]\,b[2]^2 + U^2\,V^2\,Z^2\,a[1]^2\,a[2]^2\,b[1]^2\,b[2]^2
\end{aligned}
$$

In[16]:=
```
(*The following computes the arithmetic factor \mathfrak{A}_{p} of the paper*)
```

In[17]:=
```
GL2inSym = (SymmetricReduction[#, {a[1], a[2]}, {A[1], ω}][[1]] & /@
    (SymmetricReduction[#, {b[1], b[2]}, {B[1], ω}][[1]] & /@ NormaSum[2])) /. ω^
    k_. :> ω
```

Out[17]=

$$1 - U^2 V^2 \omega - Z^2 \omega + U^2 V^2 Z^2 \omega + U V^2 \omega A[1] - V Z \omega A[1] - U^2 V Z \omega A[1] + U Z^2 \omega A[1] + U V Z \omega A[1]^2 + U^2 V \omega B[1] - U Z \omega B[1] - U V^2 Z \omega B[1] + V Z^2 \omega B[1] - U V A[1] \times B[1] - U V Z^2 \omega A[1] \times B[1] + U V Z \omega B[1]^2$$

In[18]:=
```
Collect[GL2inSym,
 DeleteCases[Times @@ ({A[1], B[1]}^#) & /@ Tuples[Range[0, 2], 2], 1]]
```

Out[18]=

$$1 - U^2 V^2 \omega - Z^2 \omega + U^2 V^2 Z^2 \omega + \left(U V^2 \omega - V Z \omega - U^2 V Z \omega + U Z^2 \omega\right) A[1] + U V Z \omega A[1]^2 + \left(U^2 V \omega - U Z \omega - U V^2 Z \omega + V Z^2 \omega\right) B[1] + \left(-U V - U V Z^2 \omega\right) A[1] \times B[1] + U V Z \omega B[1]^2$$

In[19]:=
```
Factor[(U V^2 ω - V Z ω - U^2 V Z ω + U Z^2 ω) A[1] + (U^2 V ω - U Z ω - U V^2 Z ω + V Z^2 ω) B[1]]
```

Out[19]=

$$-((U V - Z)\,\omega\,(-V A[1] + U Z A[1] - U B[1] + V Z B[1]))$$

In[20]:=
```
(*Unramified bounding*)
```

In[21]:=
```
UnrListMono2 = List @@ (Expand[GL2inSym /. ω → 1]);
```

In[22]:=
```
UnrListexpo2 =
  Exponent[#, p] & /@ (List @@ UnrListMono2 /. {A[1] → p^Th, B[1] → p^Th,
       U → p^(-1/2), V → p^(-1/2), Z → p^(-1)} /. Th → 7/64) // N;
```

In[23]:=
```
Total[UnrListMono2[[#]] & /@ Flatten@Position[# < -1 & /@ UnrListexpo2, False]]
```

Out[23]=

$$1 - U V A[1] \times B[1]$$

In[24]:=
```
Expand[GL2inSym /. ω → 0]
```

Out[24]=

$$1 - U V A[1] \times B[1]$$

In[25]:=
```
(*PGL(3): the calculation is divided into 3 parts*)
```

In[26]:=
```
Part1Norm =
  Cancel[Sum[Schu[g, 3, a] * Cancel[FF2[3, g] * Symb3 * InvL[V, 3, b]] * Z^g,
     {g, 0, Infinity}] * InvRSL[Z, 3, a, b]];
```

In[27]:=
```
Length[Part1Norm]
```

Out[27]=

360

In[28]:=
```
Part1Symm =
 Collect[(SymmetricReduction[#, {a[1], a[2], a[3]}, {A[1], A[2], ω}][[1]] & /@
     (SymmetricReduction[#, {b[1], b[2], b[3]}, {B[1], B[2], ω}][[1]] & /@
       Cancel[Part1Norm / Symb3])) /. ω^k_. :> ω,
  DeleteCases[Times @@ ({A[1], B[1], A[2], B[2]}^#) & /@ Tuples[Range[0, 2], 4], 1]]
```

Out[28]=

$$1 - 2 Z^3 \omega + Z^6 \omega + \left(V^2 Z \omega - V^2 Z^4 \omega\right) A[1] + V Z^2 \omega A[1]^2 + \left(-V Z^2 \omega + V Z^5 \omega\right) A[2] + Z^3 \omega A[1] \times A[2] + V^2 Z^4 \omega A[2]^2 + \left(V Z^3 \omega - V Z^6 \omega\right) B[1] - Z^4 \omega A[1] \times B[1] + \left(-V^2 Z^2 \omega - V^2 Z^5 \omega - V Z^3 \omega A[1]\right) A[2] \times B[1] + V^2 Z^3 \omega B[1]^2 + V Z^4 \omega A[1] B[1]^2 + \left(-V^2 Z^3 \omega + V^2 Z^6 \omega\right) B[2] + \left(-V Z - V Z^4 \omega\right) A[1] \times B[2] + Z^3 \omega B[1] \times B[2] - V Z^3 \omega B[1]^2 B[2] + A[2] \left(-Z^2 + V Z^2 B[1]\right) B[2] + V Z^3 \omega B[2]^2$$

In[29]:=
```
Part2Norm =
  Cancel[Sum[Schu[g, 3, b] * Cancel[FF1[3, g] * Syma3 * InvL[U, 3, a]] * Z^g,
     {g, 0, Infinity}] * InvRSL[Z, 3, a, b]];
```

In[30]:=
```
Length[Part2Norm]
```

Out[30]=

360

In[31]:=
```
Part2Symm =
 Collect[(SymmetricReduction[#, {a[1], a[2], a[3]}, {A[1], A[2], ω}][[1]] & /@
     (SymmetricReduction[#, {b[1], b[2], b[3]}, {B[1], B[2], ω}][[1]] & /@
       Cancel[Part2Norm / Syma3])) /. ω^k_. :> ω,
  DeleteCases[Times @@ ({A[1], B[1], A[2], B[2]}^#) & /@ Tuples[Range[0, 2], 4], 1]]
```

Out[31]=

$$1 - 2 Z^3 \omega + Z^6 \omega + \left(U Z^3 \omega - U Z^6 \omega\right) A[1] + U^2 Z^3 \omega A[1]^2 + \left(-U^2 Z^3 \omega + U^2 Z^6 \omega\right) A[2] + Z^3 \omega A[1] \times A[2] - U Z^3 \omega A[1]^2 A[2] + U Z^3 \omega A[2]^2 + \left(U^2 Z \omega - U^2 Z^4 \omega\right) B[1] - Z^4 \omega A[1] \times B[1] + U Z^4 \omega A[1]^2 B[1] + \left(-U Z - U Z^4 \omega\right) A[2] \times B[1] + U Z^2 \omega B[1]^2 + \left(-U Z^2 \omega + U Z^5 \omega\right) B[2] + \left(-U^2 Z^2 \omega - U^2 Z^5 \omega\right) A[1] \times B[2] + \left(-Z^2 + U Z^2 A[1]\right) A[2] \times B[2] + \left(Z^3 \omega - U Z^3 \omega A[1]\right) B[1] \times B[2] + U^2 Z^4 \omega B[2]^2$$

In[32]:=
```
Part3Norm = Cancel[
   Sum[Schu[g, 3, a] * Schu[g, 3, b] * Z^g, {g, 0, Infinity}] * InvRSL[Z, 3, a, b]];
```

In[33]:=
```
Length[Part3Norm]
```

Out[33]=

33

In[34]:=
```
Part3Symm =
 Collect[(SymmetricReduction[#, {a[1], a[2], a[3]}, {A[1], A[2], ω}][[1]] & /@
     (SymmetricReduction[#, {b[1], b[2], b[3]}, {B[1], B[2], ω}][[1]] & /@
       Part3Norm)) /. ω^k_. :> ω, DeleteCases[
   Times @@ ({A[1], B[1], A[2], B[2]}^#) & /@ Tuples[Range[0, 2], 4], 1]]
```

Out[34]=

$$1 - 2 Z^3 \omega + Z^6 \omega + Z^3 \omega A[1] \times A[2] - Z^4 \omega A[1] \times B[1] - Z^2 A[2] \times B[2] + Z^3 \omega B[1] \times B[2]$$

In[35]:=
```
InvLSymm3a =
  SymmetricReduction[InvL[U, 3, a], {a[1], a[2], a[3]}, {A[1], A[2], ω}][[1]]
```
Out[35]=
```
1 - U^3 ω - U A[1] + U^2 A[2]
```

In[36]:=
```
InvLSymm3b =
  SymmetricReduction[InvL[V, 3, b], {b[1], b[2], b[3]}, {B[1], B[2], ω}][[1]]
```
Out[36]=
```
1 - V^3 ω - V B[1] + V^2 B[2]
```

In[37]:=
```
GL3inSym = Expand[FullSimplify[(Part1Symm * InvLSymm3a + Part2Symm * InvLSymm3b -
       Part3Symm * InvLSymm3a * InvLSymm3b)]] /. {ω^k_. :> ω};
```

In[38]:=
```
(*Ramified Case*)
```

In[39]:=
```
RamListMono3 = List @@ (GL3inSym /. ω → 0)
```
Out[39]=
```
{1, -U V A[1] × B[1], U^2 V A[2] × B[1], -U Z A[2] × B[1], U V Z A[2] B[1]^2, U V^2 A[1] × B[2],
 -V Z A[1] × B[2], U V Z A[1]^2 B[2], -U^2 V^2 A[2] × B[2], -Z^2 A[2] × B[2],
 -U^2 V Z A[1] × A[2] × B[2], U Z^2 A[1] × A[2] × B[2], -U V^2 Z A[2] × B[1] × B[2],
 V Z^2 A[2] × B[1] × B[2], -U V Z^2 A[1] × A[2] × B[1] × B[2], U^2 V^2 Z^2 A[2]^2 B[2]^2}
```

In[40]:=
```
Length[RamListMono3]
```
Out[40]=
```
16
```

In[41]:=
```
RamListexpo3 =
  Exponent[#, p] & /@ (List @@ RamListMono3 /. {A[n_] :> p^(n * Th), B[n_] :> p^(n * Th),
       U → p^(-1/2), V → p^(-1/2), Z → p^(-1)} /. Th → 5/14) // N;
```

In[42]:=
```
Total[RamListMono3[[#]] & /@ Flatten@Position[# < 0 & /@ RamListexpo3, False]]
```
Out[42]=
```
1
```

In[43]:=
```
(*Unramified Case*)
```

In[44]:=
```
UnrListMono3 = List @@ (Expand[GL3inSym /. ω → 1]);
```

In[45]:=
```
Length[UnrListMono3]
```
Out[45]=
```
158
```

In[46]:=
```
UnrListexpo3 =
  Exponent[#, p] & /@ (List @@ UnrListMono3 /. {A[n_] :> p^Th, B[n_] :> p^Th,
       U → p^(-1/2), V → p^(-1/2), Z → p^(-1)} /. Th → 5/14) // N;
```

In[47]:=
```
(*The part of the arithmetic factor \mathfrak
  {A}_{p} that can't be bounded by Kim-Sarnak alone*)
```

In[48]:=
```
Total[UnrListMono3[[#]] & /@ Flatten@Position[# < -1 & /@ UnrListexpo3, False]]
```
Out[48]=
```
1 - U V A[1] × B[1] + U^2 V A[2] × B[1] - U Z A[2] × B[1] +
 U V Z A[2] B[1]^2 + U V^2 A[1] × B[2] - V Z A[1] × B[2] + U V Z A[1]^2 B[2]
```