```
(*This notebook computes the ratios of PGL3 L-functions for mollification*)

In[1]:= SetDirectory[NotebookDirectory[]];
        << gln.m

GL(n)pack version 15th April 2015 loading ....Copyright(c) 2002-2006
  Kevin A. Broughan  GL(n)pack comes with ABSOLUTELY NO WARRANTY. This is
  free software, and you are welcome to redistribute it under the terms
  of the GNU Public Licence. For assistance consult kab@waikato.ac.nz .

SetDelayed: Tag UpperTriangularMatrixQ in UpperTriangularMatrixQ[m_] is Protected.

GL(n)PACK loaded.

In[3]:= del = 1 / 100 000;

In[4]:= SubRules = Join[Thread[{A1, B1} → p ^ Th], Thread[{x, y, z, z1, z2} → p ^ (-1 + del)]];

In[5]:= A[n_] := SchurPolynomial[{a[1], a[2], a[3]}, {0, n}]

In[6]:= (*B stands for dual*)

In[7]:= B[n_] := SchurPolynomial[{1 / a[1], 1 / a[2], 1 / a[3]}, {0, n}]

In[8]:= muA = {-A[1], B[1], -1};

In[9]:= muB = {-B[1], A[1], -1};

In[10]:= RS[x_] := Product[1 - (a[i] / a[j]) x, {i, 1, 3}, {j, 1, 3}]

In[11]:= (*The following is  \mathcal{L}^{(d)}(x;\Pi) in the paper*)

In[12]:= L[d_, x_] := Cancel[ Sum[A[n + d] * B[n] * x ^ n, {n, 0, Infinity}] * RS[x]]

In[13]:= (*The following is  \mathfrak{U}^{(d)}(y;\Pi) *)

In[14]:= MuPoly[d_, y_] := muB[[d]] + Sum[muA[[h]] * muB[[d + h]] * y ^ h, {h, 1, 3 - d}]

In[15]:=

In[16]:= (*Computing z_{1}^d \mathfrak{U}^{(d)}(y;\widetilde{\Pi})\mathcal{L}^{(d)}(x;\Pi)*)

In[17]:= Polybefcan =
           Table[Assuming[a[1] * a[2] * a[3] == 1, z ^ n * MuPoly[n, y] * L[n, x] // FullSimplify], {n, 1, 3}];

In[18]:= FacCan = {-z (1 - x) ^ 2 (1 + x), (-1 + x) ^ 2 (1 + y) z ^ 2, -(-1 + x) ^ 2 z ^ 3};

In[19]:= Polycancrawd = Table[Cancel[Polybefcan[[n]] / FacCan[[n]]], {n, 1, 3}];

In[20]:= Polyd =
          Table[SymmetricReduction[Polycancrawd[[n]], {a[1], a[2], a[3]}, {A1, B1, 1}][[1]] // Factor, {n, 1, 3}]

Out[20]=
```

$$\{(A1 + A1\,x - B1^2\,x + A1\,x^2)\,(B1 + A1^2\,y + B1\,y^2),$$
$$A1\,(A1^2 - B1 + 2\,A1^2\,x - B1\,x - A1\,B1^2\,x + A1^2\,x^2 - B1\,x^2),\ 1 + A1^3 - 2\,A1\,B1 + 2\,x + 2\,A1^3\,x -$$
$$3\,A1\,B1\,x - A1^2\,B1^2\,x + B1^3\,x + 3\,x^2 + A1^3\,x^2 - 4\,A1\,B1\,x^2 + B1^3\,x^2 + 2\,x^3 - A1\,B1\,x^3 + x^4\}$$

In[21]:=
```
Table[Collect[Polyd[[n]], x], {n, 2, 3}]
```

Out[21]=

$$\{A1\,(A1^2 - B1) + A1\,(2\,A1^2 - B1 - A1\,B1^2)\,x + A1\,(A1^2 - B1)\,x^2,$$
$$1 + A1^3 - 2\,A1\,B1 + (2 + 2\,A1^3 - 3\,A1\,B1 - A1^2\,B1^2 + B1^3)\,x + (3 + A1^3 - 4\,A1\,B1 + B1^3)\,x^2 + (2 - A1\,B1)\,x^3 + x^4\}$$

In[22]:=
```
MonoList = Table[FacCan[[n]] * Polyd[[n]] // ExpandAll // MonomialList, {n, 1, 3}];
```

In[23]:=
```
MonoListExpo = Table[Exponent[#, p] & /@ (MonoList[[n]] /. SubRules /. Th → 5 / 14) // N, {n, 1, 3}];
```

In[24]:=
```
FirstKS =
  Table[MonoList[[n]][[Flatten@Position[# < -1 & /@ MonoListExpo[[n]], False]]] /. {z → z1}, {n, 1, 3}] //
   Flatten
```

Out[24]=

$$\{-A1^3\,y\,z1,\ -A1\,B1\,z1,\ B1^3\,x\,z1,\ A1^3\,z1^2\}$$

In[25]:=
```
DualFirstKS = FirstKS /. {A1 → B1, B1 → A1, z1 → z2}
```

Out[25]=

$$\{-B1^3\,y\,z2,\ -A1\,B1\,z2,\ A1^3\,x\,z2,\ B1^3\,z2^2\}$$

In[26]:=
```
(*The following is the polynomial \mathbf{P}*)
```

In[27]:=
```
Storemypoly = 1 + A1 B1 y + Total[FirstKS] + Total[DualFirstKS]
```

Out[27]=

$$1 + A1\,B1\,y - A1\,B1\,z1 + B1^3\,x\,z1 - A1^3\,y\,z1 + A1^3\,z1^2 - A1\,B1\,z2 + A1^3\,x\,z2 - B1^3\,y\,z2 + B1^3\,z2^2$$

In[28]:=
```
(*Comparison with other 3 Rankin-Selberg*)
```

In[29]:=
```
RSRatio3 = RS[y] ^ (-1) * RS[z1] * RS[z2];
```

In[30]:=
```
RSExpMono =
  Table[SymmetricReduction[Assuming[a[1] × a[2] × a[3] == 1, Coefficient[RS[y], y, n] // Together //
       FullSimplify], {a[1], a[2], a[3]}, {A1, B1, 1}][[1]], {n, 0, 3}]
```

Out[30]=

$$\{1,\ -A1\,B1,\ A1^3 - 2\,A1\,B1 + B1^3,\ -3 - A1^3 + 6\,A1\,B1 - A1^2\,B1^2 - B1^3\}$$

In[31]:=
```
RSinvpol[y_] := Sum[RSExpMono[[n]] * y ^ (n - 1), {n, 1, 3}]
```

In[32]:=
```
RSinvpol[y]
```

Out[32]=

$$1 - A1\,B1\,y + (A1^3 - 2\,A1\,B1 + B1^3)\,y^2$$

In[33]:=
```
RScanList = Storemypoly * RSinvpol[y] // ExpandAll // MonomialList;
```

In[34]:=
```
Listexpocance = Exponent[#, p] & /@ (RScanList /. SubRules /. Th → 5 / 14) // N;
```

In[35]:=
```
Aftercleany = Total[RScanList[[#]] & /@ Flatten@Position[# < -1 & /@ Listexpocance, False]]
```

Out[35]=

$$1 + A1^3\,y^2 - A1^2\,B1^2\,y^2 + B1^3\,y^2 - A1\,B1\,z1 + B1^3\,x\,z1 - A1^3\,y\,z1 +$$
$$A1^2\,B1^2\,y\,z1 + A1^3\,z1^2 - A1\,B1\,z2 + A1^3\,x\,z2 + A1^2\,B1^2\,y\,z2 - B1^3\,y\,z2 + B1^3\,z2^2$$

In[36]:=
```
(*Next,let's clean with 1/(RS) over z1 z2*)
```

In[37]:=
```
RSusuMono = Table[SymmetricReduction[Assuming[a[1] × a[2] × a[3] == 1,
      Together[Coefficient[Series[RS[z]^(-1), {z, 0, 4}], z, n]] // FullSimplify],
    {a[1], a[2], a[3]}, {A1, B1, 1}][[1]], {n, 0, 4}];
```

In[38]:=
```
RSusual[z_] := Sum[RSusuMono[[n]] * z^(n - 1), {n, 1, 4}]
```

In[39]:=
```
RSusual[z]
```

Out[39]=

$$1 + A1\,B1\,z + \left(-A1^3 + 2\,A1\,B1 + A1^2\,B1^2 - B1^3\right) z^2 + \left(3 + A1^3 - 6\,A1\,B1 - 2\,A1^4\,B1 + 5\,A1^2\,B1^2 + B1^3 + A1^3\,B1^3 - 2\,A1\,B1^4\right) z^3$$

In[40]:=
```
FinalList = Aftercleany * RSusual[z1] * RSusual[z2] // ExpandAll // MonomialList;
```

In[41]:=
```
FinalListexpocance = Exponent[#, p] & /@ (FinalList /. SubRules /. Th → 5 / 14) // N;
```

In[42]:=
```
(*The following outputs the polynomial \mathbf{Q}*)
```

In[43]:=
```
Total[FinalList[[#]] & /@ Flatten@Position[# < -1 & /@ FinalListexpocance, False]]
```

Out[43]=

$$1 + A1^3\,y^2 - A1^2\,B1^2\,y^2 + B1^3\,y^2 + B1^3\,x\,z1 - A1^3\,y\,z1 + A1^2\,B1^2\,y\,z1 - A1^3\,B1^3\,y^2\,z1 - B1^3\,z1^2 + A1^3\,B1^3\,y\,z1^2 + A1^3\,x\,z2 + A1^2\,B1^2\,y\,z2 - B1^3\,y\,z2 - A1^3\,B1^3\,y^2\,z2 - A1^2\,B1^2\,z1\,z2 + 2\,A1^3\,B1^3\,y\,z1\,z2 - A1^3\,B1^3\,z1^2\,z2 - A1^3\,z2^2 + A1^3\,B1^3\,y\,z2^2 - A1^3\,B1^3\,z1\,z2^2$$