```
(*This notebook computes the finite Euler
 product over p \mid \mathfrak{k} in the U^(r) sum.*)

In[1]:= SetDirectory[NotebookDirectory[]];
<< gln.m

GL(n)pack version 15th April 2015 loading ....Copyright(c) 2002-2006
  Kevin A. Broughan  GL(n)pack comes with ABSOLUTELY NO WARRANTY. This is
  free software, and you are welcome to redistribute it under the terms
  of the GNU Public Licence. For assistance consult kab@waikato.ac.nz .

SetDelayed: Tag UpperTriangularMatrixQ in UpperTriangularMatrixQ[m_] is Protected.

GL(n)PACK loaded.

In[3]:= Schu[n_, d_, nota_ : a] :=
 SchurPolynomial[Table[nota[i], {i, 1, d}], Join[ConstantArray[0, d - 2], {n}]]

In[4]:= (*U= psi(p)/p^u and Z=omega/p^z*)

In[5]:= Syma2 = a[1] - a[2];

In[6]:= Syma3 = (a[1] - a[2]) (a[1] - a[3]) (a[2] - a[3]);

In[7]:= InvL[X_, n_] := Product[(1 - a[i] X), {i, 1, n}]

In[8]:= assu = Element[k, Integers] && k ≥ 2;

In[9]:= coeffsAndMonoms[p_, vars_] :=
 Module[{rules, exps, coeffs, monoms}, rules = CoefficientRules[p, vars];
  exps = Keys[rules];
  coeffs = Values[rules];
  monoms = Times @@ MapThread[Power, {vars, #}] & /@ exps;
  {coeffs, monoms}]

In[10]:= commonMonomialFactor[expr_, vars_, ass_ : True] :=
 Module[{terms, exps, mins}, terms = List @@ Expand[expr];
  exps = Table[Exponent[t, v], {t, terms}, {v, vars}];
  mins = Simplify[Min /@ Transpose[exps], ass];
  Times @@ MapThread[#1 ^ #2 &, {vars, mins}]]

In[11]:= pullCommonMonomial[expr_, vars_, ass_ : True] :=
 Module[{fac}, fac = commonMonomialFactor[expr, vars, ass];
  fac FullSimplify[expr / fac, ass]]

In[12]:= (*The case for PGL(2)*)

In[13]:= Sortsum2 =
 Collect[InvL[U, 2] * InvL[Z, 2] * Syma2 * Sum[Sum[Schu[M + g + k, 2, a] * U ^ M, {M, 0, Infinity}] * Z ^ g,
     {g, 0, Infinity}] // FullSimplify // Expand, {U, Z, U * Z}]

Out[13]=
```

$$a[1]^{1+k} - a[2]^{1+k} + U\left(-a[1]^{1+k}\, a[2] + a[1]\, a[2]^{1+k}\right) + Z\left(-a[1]^{1+k}\, a[2] + a[1]\, a[2]^{1+k}\right) + U Z\left(a[1]^{1+k}\, a[2]^{2} - a[1]^{2}\, a[2]^{1+k}\right)$$

```
In[14]:= GL2CoefinHec =
  pullCommonMonomial[#, {a[1], a[2]}, assu] & /@ coeffsAndMonoms[Sortsum2, {U, Z}][[1]] /.
   {a[1]^(k-·) - a[2]^(k-·) :> Hec[k - 1], -a[1]^(k-·) + a[2]^(k-·) :> -Hec[k - 1]}

Out[14]=
{a[1]^2 a[2]^2 Hec[-2 + k], -a[1] × a[2] × Hec[-1 + k], -a[1] × a[2] × Hec[-1 + k], Hec[k]}

In[15]:= (*The following gives \mathfrak{E}_{p}(U,Z;\Pi;k) in the paper*)

In[16]:= Total[GL2CoefinHec * coeffsAndMonoms[Sortsum2, {U, Z}][[2]]] /. {a[1]^k_. × a[2]^k_. :> ω}

Out[16]=
U Z ω Hec[-2 + k] - U ω Hec[-1 + k] - Z ω Hec[-1 + k] + Hec[k]

In[17]:= (*The case of PGL(3)*)

In[18]:= Sortsum3 = Collect[InvL[U, 3] * InvL[Z, 3] * Syma3 *
      Sum[Sum[Schu[M + g + k, 3, a] * U^M, {M, 0, Infinity}] * Z^g, {g, 0, Infinity}] //
     FullSimplify // Expand, {U, U^2, Z, Z^2, U*Z, U^2*Z, U*Z^2, U^2*Z^2}];

In[19]:= Rulpl = {a[1] :> Hec[1], a[1]^k_. - a[2]^k_. :> Hec[k - 1],
    a[1]^(k + n_.) × a[2]^r_. - a[1]^(k + n_.) × a[3]^r_. + a[2]^(k + n_.) × a[3]^r_. - a[2]^(k + n_.) × a[1]^r_. +
     a[3]^(k + n_.) × a[1]^r_. - a[3]^(k + n_.) × a[2]^r_. :> Fou[r - 1, k + n - r - 1]};

In[20]:= Rulmin = {a[1] :> Hec[1], -a[1]^k_. + a[2]^k_. :> -Hec[k - 1],
    -a[1]^(k + n_.) × a[2]^r_. + a[1]^(k + n_.) × a[3]^r_. - a[2]^(k + n_.) × a[3]^r_. + a[2]^(k + n_.) × a[1]^r_. -
     a[3]^(k + n_.) × a[1]^r_. + a[3]^(k + n_.) × a[2]^r_. :> -Fou[r - 1, k + n - r - 1]};

In[21]:= Assuming[a[1] × a[2] × a[3] == 1 && assu , Expand[FullSimplify[coeffsAndMonoms[Sortsum3, {U, Z}]]]] /.
  {Rulpl[[3]], Rulmin[[3]] }

Out[21]=
{{Fou[0, -2 + k], -Fou[1, -2 + k], Fou[0, -1 + k], -Fou[1, -2 + k],
   -a[1] a[2]^(1+k) + a[1]^(2+k) a[2]^2 a[3] + a[2]^(1+k) a[3] - a[1]^(2+k) a[2] a[3]^2 + a[1] a[3]^(1+k) - a[2] a[3]^(1+k) + Fou[2, -2 + k],
   -Fou[1, -1 + k], Fou[0, -1 + k], -Fou[1, -1 + k], Fou[0, k]}, {U^2 Z^2, U^2 Z, U^2, U Z^2, U Z, U, Z^2, Z, 1}}

In[22]:= GL3CoefinHec = pullCommonMonomial[#, {a[1], a[2], a[3]}, assu] & /@
    coeffsAndMonoms[Sortsum3, {U, Z}][[1]] /. {a[1]^k_. × a[2]^k_. × a[3]^k_. :> ω};

In[23]:= OmePos = Flatten@Position[GL3CoefinHec, p_ /; ! FreeQ[p, ω], {1}]

Out[23]=
{1, 2, 3, 4, 7}

In[24]:= coeffsAndMonoms[Sortsum3, {U, Z}][[2]][[OmePos]]

Out[24]=
{U^2 Z^2, U^2 Z, U^2, U Z^2, Z^2}
```

```
In[25]:= GL3inSchur = MapAt[ω # &, Expand[GL3CoefinHec /. ω → 1] /. {Rulpl[[3]], Rulmin[[3]]}, List /@ OmePos]

Out[25]=
{ω Fou[0, -2 + k], -ω Fou[1, -2 + k], ω Fou[0, -1 + k], -ω Fou[1, -2 + k],
  a[1]^(2+k) a[2]^2 a[3] - a[1]^2 a[2]^(2+k) a[3] - a[1]^(2+k) a[2] a[3]^2 + a[1] a[2]^(2+k) a[3]^2 + a[1]^2 a[2] a[3]^(2+k) -
   a[1] a[2]^2 a[3]^(2+k) + Fou[2, -2 + k], -Fou[1, -1 + k], ω Fou[0, -1 + k], -Fou[1, -1 + k], Fou[0, k]}

In[26]:= (*Ramified case*)

In[27]:= GL3inSchurRam = Assuming[a[1] × a[2] × a[3] == 0, FullSimplify[GL3inSchur /. {ω → 0}]]

Out[27]=
{0, 0, 0, 0, Fou[2, -2 + k], -Fou[1, -1 + k], 0, -Fou[1, -1 + k], Fou[0, k]}

In[28]:= GL3FouMonoRam = GL3inSchurRam * coeffsAndMonoms[Sortsum3, {U, Z}][[2]] /. Z → 0

Out[28]=
{0, 0, 0, 0, 0, -U Fou[1, -1 + k], 0, 0, Fou[0, k]}

In[29]:= (*This gives \mathfrak{E}_{p}(U,Z;\Pi;k) for p ram. in the paper*)

In[30]:= Total[GL3FouMonoRam]

Out[30]=
Fou[0, k] - U Fou[1, -1 + k]

In[31]:= (*Unramified case*)

In[32]:= GL3inSchurUnr = Assuming[a[1] × a[2] × a[3] == 1 && assu,
   Expand[FullSimplify[Expand[GL3CoefinHec /. ω → 1] /. {Rulpl[[3]], Rulmin[[3]]}]] /.
    {Rulpl[[3]], Rulmin[[3]]}]

Out[32]=
{Fou[0, -2 + k], -Fou[1, -2 + k], Fou[0, -1 + k], -Fou[1, -2 + k],
  Fou[0, -1 + k] + Fou[2, -2 + k], -Fou[1, -1 + k], Fou[0, -1 + k], -Fou[1, -1 + k], Fou[0, k]}

In[33]:= GL3FouMonoUnr = GL3inSchurUnr * coeffsAndMonoms[Sortsum3, {U, Z}][[2]]

Out[33]=
{U^2 Z^2 Fou[0, -2 + k], -U^2 Z Fou[1, -2 + k], U^2 Fou[0, -1 + k], -U Z^2 Fou[1, -2 + k],
  U Z (Fou[0, -1 + k] + Fou[2, -2 + k]), -U Fou[1, -1 + k], Z^2 Fou[0, -1 + k], -Z Fou[1, -1 + k], Fou[0, k]}

In[34]:= (*This gives \mathfrak{E}_{p}(U,Z;\Pi;k) for unr. p in the paper*)

In[35]:= Total[GL3FouMonoUnr]

Out[35]=
U^2 Z^2 Fou[0, -2 + k] + U^2 Fou[0, -1 + k] + Z^2 Fou[0, -1 + k] + Fou[0, k] - U^2 Z Fou[1, -2 + k] -
  U Z^2 Fou[1, -2 + k] - U Fou[1, -1 + k] - Z Fou[1, -1 + k] + U Z (Fou[0, -1 + k] + Fou[2, -2 + k])

In[36]:= (*In terms of Hecke eigenvalues*)
UnrRepla = Fou[m_., n_.] :→ dHec[m] * Hec[n] - dHec[m - 1] * Hec[n - 1];
```

In[37]:=
```
CoefUnrInHec = Coefficient[Total[GL3FouMonoUnr] /. UnrRepla, Table[Hec[k - i], {i, 0, 3}]]
```

Out[37]=

$$\{dHec[0],\ -dHec[-1] + U^2\ dHec[0] + U\ Z\ dHec[0] + Z^2\ dHec[0] - U\ dHec[1] - Z\ dHec[1],\ -U^2\ dHec[-1] - U\ Z\ dHec[-1] - Z^2\ dHec[-1] + U\ dHec[0] + Z\ dHec[0] + U^2\ Z^2\ dHec[0] - U^2\ Z\ dHec[1] - U\ Z^2\ dHec[1] + U\ Z\ dHec[2],\ -U^2\ Z^2\ dHec[-1] + U^2\ Z\ dHec[0] + U\ Z^2\ dHec[0] - U\ Z\ dHec[1]\}$$

In[38]:=
```
(*The f-function evaluates at 1,p, p^2, p^3*)
```

In[39]:=
```
farith = SymmetricReduction[#, {a[1], a[2], a[3]}, {A[1], A[2], 1}][[1]] & /@
    (CoefUnrInHec /. {dHec[m_.] :> SchurPolynomial[{a[1], a[2], a[3]}, {m, 0}]}) // Expand
```

Out[39]=

$$\{1,\ U^2 + U\ Z + Z^2 - U\ A[2] - Z\ A[2],\ U + Z + U^2\ Z^2 - U\ Z\ A[1] - U^2\ Z\ A[2] - U\ Z^2\ A[2] + U\ Z\ A[2]^2,\ U^2\ Z + U\ Z^2 - U\ Z\ A[2]\}$$

In[40]:=
```
(*Bounding*)
```

In[41]:=
```
ExpoList =
  Exponent[#, p] & /@ (List @@ # & /@ farith /. {Z -> p^(-1), U -> p^(-1/2), A[1] -> p^Th, A[2] -> p^Th})
```

Out[41]=

$$\left\{0,\ \left\{-1,\ -\frac{3}{2},\ -2,\ -\frac{1}{2} + Th,\ -1 + Th\right\},\ \left\{-\frac{1}{2},\ -1,\ -3,\ -\frac{3}{2} + Th,\ -2 + Th,\ -\frac{5}{2} + Th,\ -\frac{3}{2} + 2\ Th\right\},\ \left\{-2,\ -\frac{5}{2},\ -\frac{3}{2} + Th\right\}\right\}$$

In[42]:=
```
RespScaleMax =
  Assuming[0 < Th < 1/2, Max /@ (MapIndexed[#1/#2[[1]] &, Rest[ExpoList]]) // FullSimplify]
```

Out[42]=

$$\left\{-\frac{1}{2} + Th,\ -\frac{1}{4},\ \frac{1}{6}\left(-3 + 2\ Th\right)\right\}$$

In[43]:=
```
Assuming[5/14 < Th < 1/2, Max[RespScaleMax] // FullSimplify]
```

Out[43]=

$$-\frac{1}{2} + Th$$

In[44]:=
```
(*Checking*)
```

In[45]:=
```
BackSubUnr = GL3FouMonoUnr /. {Fou[m_., n_.] :> Syma3 * SchurPolynomial[{a[1], a[2], a[3]}, {m, n}]} //
    FullSimplify // Expand;
```

In[46]:=
```
BackSubRam = GL3FouMonoRam /. {Fou[m_., n_.] :> Syma3 * SchurPolynomial[{a[1], a[2], a[3]}, {m, n}]} //
    FullSimplify // Expand;
```

In[47]:=
```
InitUnr =
  Assuming[a[1] × a[2] × a[3] == 1, FullSimplify[coeffsAndMonoms[Sortsum3, {U, Z}][[1]] // Expand] *
   coeffsAndMonoms[Sortsum3, {U, Z}][[2]];
```

In[48]:=
```
InitRam = Assuming[a[3] == 0, FullSimplify[coeffsAndMonoms[Sortsum3, {U, Z}][[1]] // Expand] *
    coeffsAndMonoms[Sortsum3, {U, Z}][[2]] /. {Z -> 0};
```

```
In[49]:= Assuming[a[3] == 0, FullSimplify[BackSubRam - InitRam]]

Out[49]=
{0, 0, 0, 0, 0, 0, 0, 0, 0}

In[50]:= Assuming[a[1] × a[2] × a[3] == 1 && assu, FullSimplify[BackSubUnr - InitUnr]]

Out[50]=
{0, 0, 0, 0, 0, 0, 0, 0, 0}

In[51]:= Assuming[a[1] × a[2] × a[3] == 1, FullSimplify[(BackSubUnr /. k → 0) - (InitUnr /. k → 0)]]

Out[51]=
{0, 0, 0, 0, 0, 0, 0, 0, 0}

In[52]:= Assuming[a[1] × a[2] × a[3] == 1, FullSimplify[(BackSubUnr /. k → 1) - (InitUnr /. k → 1)]]

Out[52]=
{0, 0, 0, 0, 0, 0, 0, 0, 0}

In[53]:= Replsch = Fou[m_., n_.] :→ SchurPolynomial[{a[1], a[2], a[3]}, {m, n}];

In[54]:= SymmetricReduction[#, {a[1], a[2], a[3]}, {AA[1], AA[2], 1}][[1]] & /@
          Map[Total, Table[GL3FouMonoUnr /. {k → r}, {r, 0, 2}] /. Replsch]

Out[54]=
```

$$\left\{1+U^2 Z+U Z^2-U Z\,AA[2],\ U^2+U Z+Z^2+AA[1]+(-U-Z)\,AA[2],\ -2U-2Z+U^2 Z^2+3(U+Z)-U Z\,AA[1]+\left(U^2+U Z+Z^2\right)AA[1]+AA[1]^2+\left(-1-U^2 Z-U Z^2\right)AA[2]+(-U-Z)\,AA[1]\times AA[2]+U Z\,AA[2]^2\right\}$$