```
(*This notebook computes the asymptotic of an Euler factor in the diagonal term;
see Section 5 of the paper.*)

In[1]:= SetDirectory[NotebookDirectory[]];
        << gln.m

GL(n)pack version 15th April 2015 loading ....Copyright(c) 2002-2006
  Kevin A. Broughan  GL(n)pack comes with ABSOLUTELY NO WARRANTY. This is
  free software, and you are welcome to redistribute it under the terms
  of the GNU Public Licence. For assistance consult kab@waikato.ac.nz .

SetDelayed: Tag UpperTriangularMatrixQ in UpperTriangularMatrixQ[m_] is Protected.

GL(n)PACK loaded.

In[3]:= Schu[n_, d_, nota_ : a] :=
         SchurPolynomial[Table[nota[i], {i, 1, d}], Join[ConstantArray[0, d - 2], {n}]]

In[4]:= InvRSL[X_, d_, nota1_ : a, nota2_ : b] :=
         Product[Product[(1 - nota1[i] * nota2[j] X), {i, 1, d}], {j, 1, d}]

In[5]:= (*x= p^(-s)*)

In[6]:= RawH[x_, d_] := Cancel[InvRSL[x, d, a, b] * Sum[Schu[m, d, a] * Schu[m, d, b] x ^ m, {m, 0, Infinity}]]

In[7]:= RawH[x, 2]

Out[7]= 1 - x^2 a[1] × a[2] × b[1] × b[2]

In[8]:= (*y=p^(-w)→0*)

In[9]:= PolyExpan[m_, n_, d_] :=
         Series[(1 - p * y) + InvRSL[x, d, a, b] * (1 / RawH[x, d]) * ((1 - y) ^ 2 - (1 - p * y)), {x, 0, m}, {y, 0, n}]

In[10]:= Symtr2[m_, n_] := SymmetricReduction[
            SymmetricReduction[Normal[PolyExpan[m, n, 2]], {a[1], a[2]}, {A[1], 1}][[1]], {b[1], b[2]}, {B[1], 1}][[1]]

In[11]:= Symtr3[m_, n_] := SymmetricReduction[
            SymmetricReduction[Normal[PolyExpan[m, n, 3]], {a[1], a[2], a[3]}, {A[1], A[2], 1}][[1]],
            {b[1], b[2], b[3]}, {B[1], B[2], 1}][[1]]

In[12]:= pExpan2[m_, n_, w_, s_, Th_] :=
          (Symtr2[m, n] /. Thread[{A[1], B[1]} → p ^ Th]) /. {y → p ^ (-w), x → p ^ (-s)} // Expand

In[13]:= pExpan3[m_, n_, w_, s_, Th_] :=
          (Symtr3[m, n] /. Thread[{A[1], A[2], B[1], B[2]} → p ^ Th]) /. {y → p ^ (-w), x → p ^ (-s)} // Expand

In[14]:= pExpon2[m_, n_, w_, s_, Th_] :=
          DeleteCases[Exponent[#, p] & /@ (List @@ pExpan2[m, n, w, s, Th]), 0]

In[15]:= pExpon3[m_, n_, w_, s_, Th_] :=
          DeleteCases[Exponent[#, p] & /@ (List @@ pExpan3[m, n, w, s, Th]), 0]
```

In[16]:=
```
Assuming[ 7 / 64 < Th < 1 / 2 && w > 1 && s > Th , Max[pExpon2[20, 20, w, s, Th]] // FullSimplify]
```

Out[16]=
```
Max[1 - s + 2 Th - w, -w]
```

In[17]:=
```
Assuming[ 5 / 14 < Th < 1 / 2 && w > 1 && s > Th , Max[pExpon3[10, 10, w, s, Th]] // FullSimplify]
```

Out[17]=
```
Max[1 - s + 2 Th - w, -w]
```